Mathematische Methoden in er Physik

Bonus Unterlagen : Aufgaben und Lösungen

Hinweis : Diese Texte wurden mit Hilfe von WOLFRAM MATHEMATICA 12 (Wolfram Research, Inc.) erstellt.

Viele in diesen Aufgaben gestellten Probleme können mit *Mathematica* direkt durch spezielle Befehle (wie etwa Limit oder Series) gelöst werden. Da es uns hier aber auf das entsprechende Hintergrundwissen ankommt, werden wir meist ausführlicher vorgehen und erst in späteren Rechnungen oder zur Kontrolle diese Befehle verwenden.

Kapitel 12. Funktionenräume

12.1

Blättern Sie zu Kap. 5 zurück, und beantworten Sie die folgenden Fragen.

Was sind die Regeln der Vektoralgebra?
Was bedeutet lineare Unabhängigkeit?
Was passiert, wenn man ein schiefwinkeliges Basissystem wählt; wie ändern sich die Ausdrücke für Länge, Vektoraddition, inneres Produkt?

Lösungsweg

Die ersten beiden Fragen müssen Sie selbst klären.

Was passiert, wenn man ein schiefwinkeliges Basissystem wählt; wie ändern sich die Ausdrücke für Länge, Vektoraddition, inneres Produkt?

Der wichtige Punkt ist: die Basisvektoren sind nicht mehr orthogonal (wohl aber linear unabhängig, sonst wäre es ja keine Basis).

Wir bezeichnen die (auf Länge 1 normierten) Basisvektoren mit $\mathbf{e}_1$, $\mathbf{e}_2$ und $\mathbf{e}_3$. Ein beliebiger Vektor kann durch seine Komponenten in diesem Basissystem ausgedrückt werden:

$$\mathbf{A} = a_1\,\mathbf{e}_1 + a_2\,\mathbf{e}_2 + a_3\,\mathbf{e}_3$$

$$\text{mit}: A \cdot \mathbf{e}_1 = a_1 + a_2\,\mathbf{e}_2 \cdot \mathbf{e}_1 + a_3\,\mathbf{e}_3 \cdot \mathbf{e}_1$$

und entsprechend für $A \cdot \mathbf{e}_2$ und $A \cdot \mathbf{e}_3$.

Wenn wir eine symmetrische Matrix folgender Art definieren:

Matrixelemnt $U_{ik} = \mathbf{e}_1 \cdot \mathbf{e}_2$,

$$U = \begin{pmatrix} 1 & e_1 \cdot e_2 & e_1 \cdot e_3 \\ e_2 \cdot e_1 & 1 & e_2 \cdot e_3 \\ e_3 \cdot e_1 & e_3 \cdot e_2 & 1 \end{pmatrix},$$

dann ist

$$\begin{pmatrix} A \cdot \mathbf{e_1} \\ A \cdot \mathbf{e_2} \\ A \cdot \mathbf{e_3} \end{pmatrix} = U . \begin{pmatrix} a_1 \\ a_2 \\ a_3 \end{pmatrix}.$$

Wenn die Basisvektoren linear unabhängig sind (und sie sind es, da sie laut Annahme eine Basis bilden), dann ist U invertierbar und wir können die Gleichungen lösen:

$$\begin{pmatrix} a_1 \\ a_2 \\ a_3 \end{pmatrix} = U^{-1} \begin{pmatrix} A \cdot \mathbf{e_1} \\ A \cdot \mathbf{e_2} \\ A \cdot \mathbf{e_3} \end{pmatrix}$$

Im Fall einer orthogonalen Basis ist U die Einheitsmatrix und diese Gleichung wird sehr einfach: $a_i = A.\mathbf{e}_i$.

Die Summe zweier Vektoren ist wie gewohnt komponentenweise anzuschreiben:

$$(A + B)_i = A_i + B_i .$$

Da Skalarprodukt von zwei Vektoren A und B wird

$$\mathbf{A} \cdot \mathbf{B} = (a_1\, \mathbf{e}_1 + a_2\, \mathbf{e}_2 + a_3\, \mathbf{e}_3) \cdot (b_1\, \mathbf{e}_1 + b_2\, \mathbf{e}_2 + b_3\, \mathbf{e}_3)$$
$$= \sum_{i,k=1}^{3} a_i\, b_k\, U_{ik} .$$

Das Längenquadrat eines Vektors ist daher

$$A^2 = (a_1\, \mathbf{e}_1 + a_2\, \mathbf{e}_2 + a_3\, \mathbf{e}_3) \cdot (a_1\, \mathbf{e}_1 + a_2\, \mathbf{e}_2 + a_3\, \mathbf{e}_3)$$
$$= \sum_{i,k=1}^{3} a_i\, a_k\, U_{ik} .$$

In[∘]:=
```
ClearAll["Global`*"];
```

12.5

Leiten Sie eine Beziehung zwischen d(f,g) und (f,g) her; verdeutlichen Sie sich diese Beziehung anhand einer Skizze für den $\mathbb{R}^2$.

Lösungsweg

Wir nehmen an, es handelt sich um den Vektorraum L^2 (vgl. S. 358). Der Ausdruck d(f,g) bezeichnet einen Abstand zwischen den beiden Elementen f und g. Auf S 358 wird gezeigt, dass d(f,g) als Norm ||f-g|| geschrieben werden kann, und diese wiederum durch ein Skalarprodukt ausgedrückt werden kann:

$$(d\,(f,\ g))^2 = (||f - g||)^2 = (f - g,\ f - g)$$

Wir formen weiter um:

$(f-g,\ f-g) = (f,\ f) - (g,\ f) - (f,\ g) + (g,\ g)$
$= ||f||\,^2 + ||g||\,^2 - 2\,\mathrm{Re}[(f,\ g)]$

Die gesuchte Beziehung ist also

$(d\ (f,\ g))^2 == ||f||\,^2 + ||g||\,^2 - 2\,\mathrm{Re}[(f,\ g)]$

Im R^2 können wir uns f und g als Seiten eines Dreiecks vorstellen. Die Seitenlängen sind :

$a = ||f||$, $b = ||g||$, und c

Das Skalarprodukt kann auch in folgender Form geschrieben werden:

$(f,\ g) = ||f||\ \ ||g||\ \cos\varphi = a\,b\cos\varphi$,

wobei φ den Winkel zwischen den Seiten f und g bezeichnet.

Damit erkennen wir, dass die obige Beziehung der so genannte Kosinus-Satz ist:

$c^2 = a^2 + b^2 - 2\,a\,b\cos\varphi$

```
In[•]:= ClearAll["Global`*"];
```

12.7

Die Menge B = { $\boldsymbol{b}_1$, $\boldsymbol{b}_2$, $\boldsymbol{b}_3$ } ist eine Basis. Bilden auch die drei Linearkombinationen

$\mathbf{a}_1 = 2\,\mathbf{b}_1 + 3\,\mathbf{b}_2 - \mathbf{b}_3$,
$\mathbf{a}_3 = \mathbf{b}_1 - 2\,\mathbf{b}_2 + 2\,\mathbf{b}_3$,
$\mathbf{a}_3 = -2\,\mathbf{b}_1 + \mathbf{b}_2 - 2\,\mathbf{b}_3$

linear unabhängige Elemente einer Basis? Der Vektor f hat im System B die Komponenten (3,-1,2), welche Komponenten hat er im System A?

Lösungsweg

Wir denken und die Basiselemente zu einem dreikomponentigen Vektor a zusammengefasst, analog b.In Matrixschreibweise ist die Beziehung zwischen a und b dann:

$$\mathbf{a} = U\,\mathbf{b}\ \text{ oder }\ \mathbf{a}_i = \sum_{k=1}^{3} U_{ik}\,\mathbf{b}_k$$

mit der Matrix

$$\text{In[•]:=}\ \mathbf{U} = \begin{pmatrix} 2 & 3 & -1 \\ 1 & -2 & 2 \\ -2 & 1 & -2 \end{pmatrix};$$

Um diese Beziehung umzukehren, also um b durch a auszudrücken, müssen wir U invertieren.

$$\mathbf{b} = U^{-1}\,\mathbf{a}\ \text{ oder }\ \mathbf{b}_i = \sum_{k=1}^{3} \left(U^{-1}\right)_{ik}\,\mathbf{a}_k\ .$$

Das geht immer, wenn die Zeilen und Spalten(also die Basisvektoren) linear unabhängig sind.

```
In[•]:= UT = Inverse[U]

Out[•]= {{2, 5, 4}, {-2, -6, -5}, {-3, -8, -7}}
```

$$= \begin{pmatrix} 2 & 5 & 4 \\ -2 & -6 & -5 \\ -3 & -8 & -7 \end{pmatrix}$$

Der Vektor f hat im B-System die Komponenten (3,-1,2) und daher:

$$\mathbf{f} = \sum_{i=1}^{3} f_i \, \mathbf{b}_i = \sum_{i,k=1}^{3} f_i \left(U^{-1}\right)_{ik} \mathbf{a}_k = \sum_{i,k=1}^{3} f'_k \, \mathbf{a}_k$$

Die Komponenten des Vektors im A-System sind also

$$f'_k = \sum_{i=1}^{3} f_i \left(U^{-1}\right)_{ik} = \sum_{i=1}^{3} \left(\left(U^{-1}\right)^T\right)_{ki} f_i$$

(Mit X^T bezeichnent wir die transponierte X - Matrix.)

Die Komponenten von f im A-System sind also

```
In[•]:= Transpose[Inverse[U]].{3, -1, 2}
Out[•]= {2, 5, 3}
```

Wir überprüfen das noch einmal durch einsetzen:

```
In[•]:= f = Expand[2 a1 + 5 a2 + 3 a3 /.
            {a1 -> 2 b1 + 3 b2 - b3,
                a2 -> b1 - 2 b2 + 2 b3,
                a3 -> -2 b1 + b2 - 2 b3}]
Out[•]= 3 b1 - b2 + 2 b3
```

Es stimmt also.

Wie geht das in die andere Richtung?
Der Vektor g habe im A-System die Komponenten (g_i) und daher:

$$\mathbf{g} = \sum_{i=1}^{3} g_i \, \mathbf{a}_i = \sum_{i,k=1}^{3} g_i \, U_{ik} \, \mathbf{b}_k = \sum_{i,k=1}^{3} g'_k \, \mathbf{b}_k$$

Die Komponenten des Vektors im B-System sind also

$$g'_k = \sum_{i=1}^{3} g_i \, U_{ik} = \sum_{i=1}^{3} \left(U^T\right)_{ki} g_i .$$

```
In[•]:= ClearAll["Global`*"];
```

12.11

Wenn f_1(x), f_2(x), f_3(x),... alle aus L^2 sind, zeigen Sie, dass für beliebige Konstanten c_1, , c_3,... $\in \mathbb{R}^2$ die Summe $c_1 f_1$(x) + c_3 f_2(x) + c_3 f_3(x) +...$\in L^2$ ist.

Lösungsweg

Wir nehmen an, dass es sich um reelle Funktionen handelt. (Der allgemeinere Fall benötigt nur etwas mehr Schreibarbeit.)
Damit die Summe

$$h = \sum_{n=1}^{N} c_n f_n$$

aus L^2 ist, muss gelten (h,h)<∞, also

$$\sum_{n,k=1}^{N} c_n c_k (f_n, f_k) \text{ beschränkt.}$$

Das gilt sicher, falls (f_n, f_k) beschränkt ist.
Für n = k gilt das wegen der Ausgangsannahme automatisch.

Für n≠k kann man dies mit Hilfe der Cauchy-Schwarz-Ungleichung (vgl. M.12.3) zeigen:

$(f_n, f_k)^2 \leq (f_n, f_n)(f_k, f_k)$

(Dies ist eine Verallgemeinerung der Ungleichung für normale Vektoren:

$(\mathbf{a} \cdot \mathbf{b})^2 = (a\, b \cos\gamma)^2 \leq a^2 b^2$).

Da die rechte Seite laut Annahme beschränkt ist, ist es auch die linke und unser Beweis ist vollständig.

```
In[∘]:= ClearAll["Global`*"];
```

12.12

Überprüfen Sie, ob die folgenden Mengen von Funktionen jeweils eine orthogonale Menge bezüglich des jeweiligen Integrationsintervalls (Skalarprodukt wie in (12.19)) bilden.

(a) $\{\sin x,\ \sin 2x,\ \sin 3x,\ \sin 4x\}$
bezüglich $x \in (-\pi, \pi)$.

(b) $\{1,\ x,\ x^2,\ x^3\}$
bezüglich $x \in (0, 1)$.

Lösungsweg

(a)

Das Skalarprodukt lautet also

$$(f, g) = \int_{-\pi}^{\pi} dx\, f(x)\, g(x)$$

Wir müssten also alle Paare (f,g) der angegebenen Funktionen untersuchen, ob all die Skalarprodukte für zwei verschiedene Funktionen verschwinden.

Dies ist eine Gelegenheit, gleich das Integral für den allgemeinen Fall zu berechnen:

$$(\text{Sin}[n\,x],\ \text{Sin}[m\,x]) = \int_{-\pi}^{\pi} dx\ \text{Sin}[n\,x]\,\text{Sin}[m\,x]$$

$$= \frac{\text{Sin}[(m-n)\,\pi]}{m-n} - \frac{\text{Sin}[(m+n)\,\pi]}{m+n} = 0 \text{ falls } n \neq \pm m$$

$$= \pi \text{ falls } m = n$$

$$= -\pi \text{ falls } m = -n$$

Diese Integration kann entweder über den Zwischenschritt der Darstellung durch Exponentialfunktionen oder mit Hilfe eines Winkeladditionstheorems

$$\text{Sin}[n\,x]\,\text{Sin}[m\,x] = \frac{1}{2}(\text{Cos}[(m-n)\,x] - \text{Cos}[(m+n)\,x])$$

verlaufen.

Damit gilt für die angegebenen Funktionen sicher Orthogonalität. Das verwundert nicht, da diese ja Basiselemente für die Fourierreihe (Kap. 12 oder auch Aufgabe 12.15) sind.

(b)

Wenn wir das Skalarprodukt für diese Aufgabe allgemein berechnen, erhalten wir:

$$(x^n,\ x^m) = \int_0^1 dx\, x^{n+m} = \left(\frac{x^{n+m+1}}{n+m+1}\right)_0^1 = \frac{1}{n+m+1}.$$

Es verschwindet also für n≠m nicht, daher sind die Funktionen nicht orthogonal!

```
In[•]:= ClearAll["Global`*"];
```

12.13

Betrachten Sie den Vektorraum: {Polynome in $x \in \mathbb{R}$}und das linear unabhängige System: $\{1, x, x^2, x^3, \ldots\}$. Bestimmen Sie daraus jeweils die ersten vier orthogonalen Polynome bezüglich der Orthogonalitätsrelationen:

$$\text{(a)}\ (f,\ g) = \int_0^\infty dx\ \exp(-x)\ f(x)\ g(x);$$

$$\text{(b)}\ (f,\ g) = \int_{-\infty}^\infty dx\ \exp\left(-x^2\right)\ f(x)\ g(x);$$

$$\text{(c)}\ (f,\ g) = \frac{1}{2}\int_{-1}^1 dx\ f(x)\,g(x)\ \ (\text{Legendre}-\text{Basis}).$$

Lösungsweg

Die Lösungsmethode ist das Gram-Schmidt-Verfahren (S 361 ff). In *Mathematica* ist dies eine spezielle Prozedur. Wir werden daher nur für (a) das verfahren explizit durchführen, bei (b) und (c) das entsprechende *Mathematica* Paket verwenden.

(a)

Wir definieren das Skalarprodukt und die Norm

```
In[•]:= ScPr[f_, g_] :=
          Integrate[Expand[E^-x f g], {x, 0, ∞}];
        MyNorm[f_] := Sqrt[ScPr[f, f]];
```

Die Ausgangsfunktionen sind:

```
In[•]:= ψ1 = 1;
        ψ2 = x;
        ψ3 = x^2;
        ψ4 = x^3;
```

Die Basisfunktionen werden dann wie folgt berechnet:

1. Schritt

In[•]:= $\varphi_1 = \psi_1 / \text{MyNorm}[\psi_1]$

Out[•]= 1

2. Schritt

Zur Berechnung von φ_2 benötigt man das Skalarprodukt (φ_1, ψ_2) :

In[•]:= `ScPr[1, x]`

Out[•]= 1

Daraus folgt

$\eta_2 = \psi_2 - (\varphi_1, \psi_2)\, \varphi_1$

In[•]:= $\eta_2 = x - 1;$
$\varphi_2 = \eta_2 / \text{MyNorm}[\eta_2]$

Out[•]= $-1 + x$

3. Schritt

Es ist

$\eta_3 = \psi_3 - (\varphi_1, \psi_3)\, \varphi_1 - (\varphi_2, \psi_3)\, \varphi_2$

mit $(\varphi_1, \psi_3) =$

In[•]:= $\text{ScPr}[1, x^2]$

Out[•]= 2

und mit $(\varphi_2, \psi_3) =$

In[•]:= $\text{ScPr}[x - 1, x^2]$

Out[•]= 4

erhält man:

In[•]:= $\eta_3 = \text{Expand}[x^2 - 2 - 4\,(x - 1)];$
$\varphi_3 = \eta_3 / \text{MyNorm}[\eta_3]$

Out[•]= $\frac{1}{2}\left(2 - 4\,x + x^2\right)$

4. Schritt

Es ist

$\eta_4 = \psi_4 - (\varphi_1, \psi_4)\, \varphi_1 - (\varphi_2, \psi_4)\, \varphi_2 - (\varphi_3, \psi_4)\, \varphi_3$

mit $(\varphi_1, \psi_4) =$

In[•]:= $\text{ScPr}[1, x^3]$

Out[•]= 6

und $(\varphi_2, \psi_4) =$

In[•]:= $\text{ScPr}[x - 1, x^3]$

Out[•]= 18

und $(\varphi_3, \psi_4) =$

In[∘]:=
```
ScPr[1/2 (2 - 4 x + x^2), x^3]
```

Out[∘]= 18

erhält man:

In[∘]:=
```
η4 = Expand[x^3 - 6 - 18 (x - 1) - 18 × 1/2 (2 - 4 x + x^2)];
φ4 = η4 / MyNorm[η4]
```

Out[∘]= $\frac{1}{6}\left(-6+18\,x-9\,x^2+x^3\right)$

Damit ist die Aufgabe gelöst. Nur zu unserer Kontrolle wollen wir auch das vorgefertigte Mathematica Paket anwenden:

In[∘]:=
```
Orthogonalize[{1, x, x^2, x^3},
 Integrate[Expand[E^-x #1 #2], {x, 0, ∞}] &]
```

Out[∘]= $\left\{1,\ -1+x,\ \frac{1}{2}\left(-2-4\,(-1+x)+x^2\right),\ \frac{1}{6}\left(-6-18\,(-1+x)+x^3-9\left(-2-4\,(-1+x)+x^2\right)\right)\right\}$

Es sind dies -bis auf ein Vorzeichen -die Laguerre-Polynome (vgl. Kap. 15, (15.65)). In der Tat können wir ja auch einzelne Basiselemente mit -1 multiplizieren, ohne dass sich an der Orthogonalität was ändert. Es ist reine Konventionssache. Bei den Laguerre-Polynomen wählt man die Vorzeichen so, dass die Basiselemente bei x=0 den Wert 1 haben.

(b)

In[∘]:=
```
Orthogonalize[{1, x, x^2, x^3}, Integrate[Expand[E^-x^2 #1 #2], {x, -∞, ∞}] &]
```

Out[∘]= $\left\{\frac{1}{\pi^{1/4}},\ \frac{\sqrt{2}\,x}{\pi^{1/4}},\ \frac{\sqrt{2}\left(-\frac{1}{2}+x^2\right)}{\pi^{1/4}},\ \frac{2\left(-\frac{3\,x}{2}+x^3\right)}{\sqrt{3}\,\pi^{1/4}}\right\}$

Vergleich mit den Hermite-Polynomen H_n (auf S 457, in Kap. 15) zeigt, dass diese Basisfunktionen

$$\varphi_n[x] = \frac{H_n[x]}{\pi^{1/4}\,2^{n/2}\,\sqrt{n!}} \quad \text{für} \quad (n = 0,\ 1,\ 2,\ 3)$$

sind.

(c)

In[∘]:=
```
Orthogonalize[{1, x, x^2, x^3}, Integrate[Expand[#1 #2], {x, -1, 1}] / 2 &]
```

Out[∘]= $\left\{1,\ \sqrt{3}\,x,\ \frac{3}{2}\sqrt{5}\left(-\frac{1}{3}+x^2\right),\ \frac{5}{2}\sqrt{7}\left(-\frac{3\,x}{5}+x^3\right)\right\}$

Die Basiselemente haben daher die Form $\sqrt{2\,n+1}\ P_n(x)$, wobei die P_n die in Kap. 15 besprochenen Legendre-Polynome sind.

In[∘]:=
```
ClearAll["Global`*"];
```

12.14

Berechnen Sie die ersten drei Koeffizienten der Entwicklung von f(x)=cos(πx) in der Legendre-Basis des vorhergehenden Beispiels.Vergleichen Sie graphisch f(x) mit der Summe der ersten drei Terme der Entwicklung. Wie ist das Ergebnis im Vergleich mit der Taylorreihe von f(x)?

Lösungsweg

In M.15.1 finden Sie eine vollständigere Übersicht zur Legendre-Reihe. Die Orthogonalitätsrelation lautet

$$(f, g) = \frac{1}{2} \int_{-1}^{1} dx \; f(x)\, g(x)$$

und die Basiselemente sind Polynome:

$$P_0 = 1, \quad P_1 = x, \quad P_2 = \frac{3}{2} x^2 - \frac{1}{2}$$

Wenn man eine Funktion in dieser Basis darstellen will, lautet der Ansatz

$$f(x) = \sum_{n=0}^{\infty} c_n P_n(x)$$

und die Koeffizienten erhält man durch Bildung des Skalarprodukts mit den einzelnen Basispolynomen:

$$(f, P_k) = c_k (P_k, P_k) \to c_k = \frac{(f, P_k)}{(P_k, P_k)}$$

In unserem Fall ist, wie man leicht nachrechnet (wie gesagt, eine allgemeine Formel gibt es in M.15.1):

$$(P_0, P_0) = (1, 1) = 1,$$

$$(P_1, P_1) = (x, x) = \frac{1}{3},$$

$$(P_2, P_2) = \left(\frac{3}{2} x^2 - \frac{1}{2}, \frac{3}{2} x^2 - \frac{1}{2}\right) = \frac{1}{5.}$$

Weiters benötigen wir

$$(\text{Cos}[x], P_0) = \frac{1}{2} \int_{-1}^{1} dx \; \text{Cos}[\pi x] = \left(\frac{\text{Sin}[\pi x]}{2\pi}\right)_{-1}^{1} = 0,$$

$$(\text{Cos}[x], P_1) = \frac{1}{2} \int_{-1}^{1} dx \; x\, \text{Cos}[\pi x] = 0$$

$$(\text{Cos}[x], P_0) = \frac{1}{2} \int_{-1}^{1} dx \left(\frac{3}{2} x^2 - \frac{1}{2}\right) \text{Cos}[\pi x] = -\frac{3}{\pi^2}$$

und daher

$$c_0 = c_1 = 0; \quad c_2 = -\frac{15}{\pi^2.}$$

Damit sind die ersten Terme der Reihendarstellung

$$\text{Cos}[\pi x] = -\frac{15}{\pi^2} P_2(x) + \ldots$$

$$= -\frac{15}{\pi^2} \left(\frac{3}{2} x^2 - \frac{1}{2}\right)$$

(Oft normiert man die Legendre Polynome mit einem Faktor $1/\sqrt{(P_k, P_k)}$,
in diesem Fall wäre $c_2 = -3\sqrt{5}/\pi^2$, die Summe aber dieselbe.)

Wir vergleichen diesen Ausdruck im Entwicklungsintervall (-1,1) mit der tatsächlichen Funktion:

In[•]:= $\texttt{Plot}\left[\left\{\texttt{Cos[x Pi]}, -\frac{15}{\pi^2}\left(\frac{3}{2}x^2 - \frac{1}{2}\right)\right\}, \{x, -1, 1\}\right]$

Out[•]=


Wir sehen, dass die Näherung eine Näherung im Mittel ist. Im Vergleich dazu wären die ersten Terme einer Taylorreihe um x=0 beim Entwicklungspunkt besser, weiter weg aber schlechter. Die Taylorreihe bis zur Ordnung x^2lautet:

$$\text{Cos}[\text{Pi}\ x] = 1 - \frac{\pi^2 x^2}{2} \ldots$$

und der graphische Vergleich ergibt:

In[•]:= $\texttt{Plot}\left[\left\{\texttt{Cos[x Pi]}, 1 - \frac{\pi^2 x^2}{2}\right\}, \{x, -1, 1\}\right]$

Out[•]=


In[•]:=
```
ClearAll["Global`*"];
```

12.15

Zeigen Sie, dass die Funktionen {sin nx, cos nx ; n=0,1,2,3...} mit dem Skalarprodukt

$$(f, g) = \frac{1}{\pi}\int_{-\pi}^{\pi} dx\ f(x)\ g(x)$$

ein Orthogonalsystem bilden.

Lösungsweg

Wir müssen dazu zeigen, dass das Skalarprodukt von zwei (beliebigen, aber) verschiedenen Elementen verschwindet. Dazu betrachten wir die Integraltypen (n,k≥0)

$$\frac{1}{\pi}\int_{-\pi}^{\pi} dx\ \text{Sin}[k\,x]\ \text{Sin}[n\,x]$$

$$\frac{1}{\pi}\int_{-\pi}^{\pi} dx\ \text{Sin}[k\,x]\ \text{Cos}[n\,x]$$

$$\frac{1}{\pi}\int_{-\pi}^{\pi} dx\ \text{Cos}[k\,x]\ \text{Cos}[n\,x]$$

In Aufgabe 12.12 haben wir das erste dieser Integrale schon gelöst, und zwar mit Hilfe eines Additionstheorems für Winkelfunktion. Hier wählen wir eine andere Lösungsmethode: Ausdrücken durch Exponentialfunktionen.

Es ist

$$\frac{1}{\pi}\int_{-\pi}^{\pi} dx\ \text{Sin}[k\,x]\ \text{Sin}[n\,x] =$$

$$-\frac{1}{4\,\pi}\int_{-\pi}^{\pi} dx\ \left(E^{Ikx} - E^{-Ikx}\right)\left(E^{Inx} - E^{-Inx}\right)$$

$$= -\frac{1}{4\,\pi}\int_{-\pi}^{\pi} dx\ \left(E^{I\,(k+n)\,x} - E^{-I\,(k-n)\,x} - E^{I\,(k-n)\,x} + E^{-I\,(k+n)\,x}\right)$$

$= 0$ für $k = n = 0$ oder $k \neq n$

$= 1$ für $k = n > 0$

Man beachte dabei, dass für Winkelfunktionen dieser Art immer gilt, dass

$$\frac{1}{\pi}\int_{-\pi}^{\pi} dx\ E^{Inx} = 0 \text{ für } n \neq 0,\ \ 2 \text{ für } n = 0.$$

Entsprechend erhält man für die anderen Integrale (jeweils n,k⩾0) die Ergebnisse

$$\frac{1}{\pi}\int_{-\pi}^{\pi} dx\ \text{Sin}[k\,x]\ \text{Cos}[n\,x] = 0$$

$$\frac{1}{\pi}\int_{-\pi}^{\pi} dx\ \text{Cos}[k\,x]\ \text{Cos}[n\,x] = 2 \text{ für } n = k = 0$$

$= 1$ für $n = k > 0$

$= 0$ für $n \neq k$

Die Funktionen bilden daher tatsächlich ein Orthogonalsystem. Alle Funktionen, außer die für n=0, also 1, sind bereits normiert. Ein vollständiges Orthogonalsystem wird mit Hilfe folgender Funktionen gebildet:

$$\left\{\frac{1}{\sqrt{2}},\ \cos x,\ \cos 2\,x,\ \cos 3\,x,\ \ldots,\ \sin x,\ \sin 2\,x,\ \sin 3\,x,\ \ldots\right\}$$

Es handelt sich dabei um die Fourierentwicklung, vgl. Kap. 12.

In[∘]:=
```
ClearAll["Global`*"];
```

12.16

Drücken Sie die folgenden Funktionen durch Komponenten in geeigneten Basissystemen aus:

(a) $f(x) = 4x - 3$

in der Basis $\{\sin x, \sin 2x, \sin 3x\}$

bezüglich $x \in (-\pi, \pi)$;

(b) $f(x) = (6 \sin x + 1)^2$

in der Basis $\left\{1, x, \frac{3}{2}x^2 - \frac{1}{2}\right\}$

bezüglich $x \in (-1, 1)$ (Legendrereihe);

(c) $f(x) = (e^x + e^{-x})$

in der Basis $\left\{1, x, \frac{3}{2}x^2 - \frac{1}{2}\right\}$

bezüglich $x \in (-1, 1)$ (Legendrereihe).

Wie gut ist die Besselsche Ungleichung erfüllt?
Überprüfen Sie graphisch die Qualität der jeweiligen Darstellung!

Lösungsweg

(a)

$f(x) = 4x - 3$

in der Basis $\{\sin x, \sin 2x, \sin 3x\}$

bezüglich $x \in (-\pi, \pi)$;

Bei der Darstellung

$F = \sum_{n=1}^{N} c_n f_n$ ist $c_n = \frac{(F, f_n)}{(f_n, f_n)}$.

Dabei ist eine etwaige Normierungskonstante des Integrals gleichgültig, da sie sich bei dieser Bestimmung weghebt.

In unserem Fall ist einheitlich $(f_n, f_n) = 1$ wenn man das Skalarprodukt als

$$(f, g) = \frac{1}{\pi} \int_{-\pi}^{\pi} dx\, f(x)\, g(x)$$

definiert.

In[∘]:=
```
F[x_] = 4 x - 3;
c[n_] := Integrate[F[x] Sin[n x], {x, -Pi, Pi}] / π;
```

Wir berechnen die drei gesuchten Koeffizienten:

In[∘]:=
```
c[1]
```

Out[∘]= 8

8

In[•]:= c[2]

Out[•]= -4

-4

In[•]:= c[3]

Out[•]= $\frac{8}{3}$

$\frac{8}{3}$

Daher ist die Näherung:

In[•]:= Fapprox[x_] = 8 Sin[x] - 4 Sin[2 x] + (8 / 3) Sin[3 x]

Out[•]= $8\,\mathrm{Sin}[x] - 4\,\mathrm{Sin}[2\,x] + \frac{8}{3}\,\mathrm{Sin}[3\,x]$

Wir vergleichen die Näherung mit der ursprünglichen Funktion:

In[•]:= Plot[{F[x], Fapprox[x]}, {x, -Pi, Pi}]

Out[•]=



Das ist wohl keine gute Näherung. Der Grund liegt eigentlich auf der Hand: alle drei Funktionen gehen bei x=0 durch 0, die gegebene Funktion ist dort aber -3. Offenbar führt dies zu den Problemen.

Was sagt dazu die Besselsche Ungleichung? Falls die einzelnen Basisfunktionen auf Norm 1 normiert sind, lautet sie:

$$(f_{\mathrm{approx}},\ f_{\mathrm{approx}}) = \sum_n c_n^2 \leq (f,\ f).$$

Die linke Seite ist

In[•]:= LS = (8) ^2 + (-4) ^2 + (8 / 3) ^2

Out[•]= $\frac{784}{9}$

Der numerische Wert ist 87.11...

Das Quadrat der Norm von f ist:

In[∘]:=
```
Simplify[
    Integrate[(4 x - 3)^2, {x, -Pi, Pi}] / Pi]
```

Out[∘]= $18 + \frac{32\,\pi^2}{3}$

mit dem numerischen Wert 123.28 .

Tatsächlich ist 87.111 nicht sehr nahe bei 123.28, die Näherung also nicht sehr gut.

(b)

$f(x) = (6 \sin x + 1)^2$

in der Basis $\left\{1,\ x,\ \frac{3}{2} x^2 - \frac{1}{2}\right\}$

bezüglich $x \in (-1, 1)$ (Legendrereihe);

Bei der Darstellung

$F = \sum_{n=1}^{N} c_n f_n$ ist $c_n = \frac{(F, f_n)}{(f_n, f_n)}$.

mit $(f, g) = \frac{1}{2} \int_{-1}^{1} dx\, f(x)\, g(x)$

Dabei ist eine etwaige Normierungskonstante des Integrals eigentlich gleichgültig, da sie sich bei der Bestimmung der c_n weghebt.

In[∘]:=
```
F[x_] = (6 Sin[x] + 1)^2;
c[0] = Integrate[F[x] , {x, -1, 1}] /
        Integrate[1, {x, -1, 1}]
c[1] = Integrate[F[x] x , {x, -1, 1}] /
        Integrate[x^2, {x, -1, 1}]
c[2] = Integrate[F[x] (3/2 x^2 - 1/2) , {x, -1, 1}] /
        Integrate[ (3/2 x^2 - 1/2)^2, {x, -1, 1}]
```

Out[∘]= $\frac{1}{2}\,(38 - 18\,\mathrm{Sin}[2])$

Out[∘]= $-36\,(\mathrm{Cos}[1] - \mathrm{Sin}[1])$

Out[∘]= $-\frac{45}{4}\,(6\,\mathrm{Cos}[2] + \mathrm{Sin}[2])$

Daher ist die Näherung:

In[∘]:=
```
Fapprox[x_] = (19 - 9 Sin[2]) + 36 (Sin[1] - Cos[1]) x
  - (45 / 8) (6 Cos[2] + Sin[2]) (3 x^2 - 1);
```

Out[∘]= $19 + 36\,x\,(-\mathrm{Cos}[1] + \mathrm{Sin}[1]) - 9\,\mathrm{Sin}[2]$

Wir vergleichen die Näherung mit der ursprünglichen Funktion:

In[•]:= `Plot[{F[x], Fapprox[x]}, {x, -1, 1}]`

Out[•]=



Das scheint eine gute Näherung zu sein.

Was sagt dazu die Besselsche Ungleichung? Falls die einzelnen Basisfunktionen auf Norm 1 normiert sind, lautet sie:

$$(f_{approx}, f_{approx}) = \sum_n c_n^2 \leq (f, f).$$

Die linke Seite ist

In[•]:= `N[Integrate[Fapprox[x]^2, {x, -1, 1}] / 2]`

Out[•]= 156.176

Das Quadrat der Norm von f ist:

In[•]:= `N[Integrate[F[x]^2, {x, -1, 1}] / 2]`

Out[•]= 220.635

Die Ungleichung 219.97<220.64 ist offenbar erfüllt, und zwar ist sie ziemlich nahe einer Gleichung, die Näherung also sehr gut.

(c)

$f(x) = (e^x + e^{-x})$

in der Basis $\left\{1, x, \frac{3}{2}x^2 - \frac{1}{2}\right\}$

bezüglich $x \in (-1, 1)$ (Legendrereihe);

Bei der Darstellung

$F = \sum_{n=1}^{N} c_n f_n$ ist $c_n = \frac{(F, f_n)}{(f_n, f_n)}.$

mit $(f, g) = \frac{1}{2}\int_{-1}^{1} dx\, f(x)\, g(x)$

Dabei ist eine etwaige Normierungskonstante des Integrals eigentlich gleichgültig, da sie sich bei der Bestimmung der c_n weghebt.

In[∘]:=
```
F[x_] = E^x + E^-x;
c[0] = Integrate[F[x] , {x, -1, 1}] /
        Integrate[1, {x, -1, 1}]
c[1] = Integrate[F[x] x , {x, -1, 1}] /
        Integrate[x^2, {x, -1, 1}]
c[2] = Integrate[F[x] (3/2 x^2 - 1/2) , {x, -1, 1}] /
        Integrate[ (3/2 x^2 - 1/2)^2, {x, -1, 1}]
```

Out[∘]= $\frac{1}{2}\left(-\frac{2}{e}+2\,e\right)$

Out[∘]= 0

Out[∘]= $\frac{5\left(-7+e^2\right)}{e}$

Daher ist die Näherung:

In[∘]:=
```
Fapprox[x_] = 1/2 (-2/E + 2 E) +
5/4 (-14/E + 2 E) (3 x^2 - 1);
```

Wir vergleichen die Näherung mit der ursprünglichen Funktion:

In[∘]:=
```
Plot[{F[x], Fapprox[x]}, {x, -1, 1}]
```

Out[∘]=



Das scheint eine gute Näherung zu sein.

Was sagt dazu die Besselsche Ungleichung? Falls die einzelnen Basisfunktionen auf Norm 1 normiert sind, lautet sie:

$$(f_{approx}\,,\ f_{approx}) = \sum_n c_n^2 \leq (f,\ f).$$

Die linke Seite ist

In[∘]:=
```
N[Integrate[Fapprox[x]^2, {x, -1, 1}] / 2]
```

Out[∘]= 5.62682

Das Quadrat der Norm von f ist:

In[ ]:=
```
N[Integrate[F[x]^2, {x, -1, 1}] / 2]
```

Out[ ]=
```
5.62686
```

Wie im Fall (b) handelt es sich offenbar um eine ausgezeichnet Näherung, was auch an der Nähe der Besselschen Ungleichung zur Parsevalschen Gleichung erkennbar ist!

In[ ]:=
```
ClearAll["Global`*"];
```